

### 4.3 Arithmetic

We obtain the system of *first-order arithmetic* if we restrict quantifiers to elements of type **nat**. Recall the induction principle for natural numbers and the rules for equality  $n =_N m$  and the less-than relation  $n < m$  summarized in Section 3.10.

As a reminder, we will prove some frequently needed properties of equality. The first is reflexivity of equality.

$$\forall x \in \mathbf{nat}. x =_N x$$

We first give the informal proof, then its translation into a formal proof language.

**Proof:** The proof is by induction on  $x$ .

**Case:**  $x = \mathbf{0}$ . Then  $\mathbf{0} =_N \mathbf{0}$  by rule  $=_N I_0$ .

**Case:**  $x = \mathbf{s}(x')$ . Then

$$\begin{array}{ll} x' =_N x' & \text{by induction hypothesis} \\ \mathbf{s}(x') =_N \mathbf{s}(x') & \text{by rule } =_N I_s. \end{array}$$

□

As a formal proof in linear format:

```
[ x : nat;                % assumption
  0 = 0;                  % by =I0 (base case)
  [x' : nat, x' = x';     % assumptions
   s(x') = s(x')];      % by =Is (induction step)
  x = x ];               % by induction on x
!x:nat. x = x;          % by !I
```

We can also write out the proof term that corresponds to the proof above.

$$\begin{aligned} refl & : \forall x \in \mathbf{nat}. x =_N x \\ & = \lambda x \in \mathbf{nat}. \mathbf{rec} \ x \\ & \quad \mathbf{of} \ r(\mathbf{0}) \Rightarrow \mathbf{eq}_0 \\ & \quad | \ r(\mathbf{s}(x')) \Rightarrow \mathbf{eq}_s(r(x')) \end{aligned}$$

As a second example, we consider transitivity of equality.

$$\forall x \in \mathbf{nat}. \forall y \in \mathbf{nat}. \forall z \in \mathbf{nat}. x =_N y \supset y =_N z \supset x =_N z$$

This time we will give the proof in three forms: as an informal mathematical proof, as a formal proof in linear form, and as an equational specification proof term.

**Proof:** The proof is by induction on  $x$ . We need to distinguish subcases on  $y$  and  $z$ .

**Case:**  $x = 0$ . Then we distinguish subcases on  $y$ .

**Case:**  $y = 0$ . Then we distinguish subcases on  $z$ .

**Case:**  $z = 0$ . Then  $0 =_N 0$  by rule  $=_N I_0$ .

**Case:**  $z = s(z')$ . Then  $y =_N z$  is impossible by rule  $=_N E_{0s}$ .

**Case:**  $y = s(y')$ . Then  $x =_N y$  is impossible by rule  $=_N E_{0s}$ .

**Case:**  $x = s(x')$ . We assume the induction hypothesis

$$\forall y \in \mathbf{nat}. \forall z \in \mathbf{nat}. x' =_N y \supset y =_N z \supset x' =_N z$$

and distinguish subcases on  $y$ .

**Case:**  $y = 0$ . Then  $x =_N y$  is impossible by rule  $=_N E_{0s}$ .

**Case:**  $y = s(y')$ . Then we distinguish subcases on  $z$ .

**Case:**  $z = 0$ . Then  $y =_N z$  is impossible by rule  $=_N E_{s0}$ .

**Case:**  $z = s(z')$ . Then we assume  $s(x') =_N s(y')$  and  $s(y') =_N s(z')$  and have to show that  $s(x') =_N s(z')$ .

We continue:

$$\begin{array}{ll} x' =_N y' & \text{by rule } =_N E_{ss} \\ y' =_N z' & \text{by rule } =_N E_{ss} \\ x' =_N z' & \text{by universal and implication eliminations} \\ & \text{from induction hypothesis} \\ s(x') =_N s(z') & \text{by rule } =_N I_s \end{array}$$

□

The formal proof of transitivity is a good illustration why mathematical proofs are not written as natural deductions: the granularity of the steps is too small even for relatively simple proofs.<sup>2</sup>

```
[ x : nat;
  [ y : nat;
    [ z : nat;
      [ 0 = 0;
        [ 0 = 0;
          0 = 0 ]]; % eqI0
        0 = 0 => 0 = 0 ];
      0 = 0 => 0 = 0 => 0 = 0; % case (z = 0)
      [ z' : nat, 0 = 0 => 0 = s(z') => 0 = s(z');
        [ 0 = 0;
          [ 0 = s(z');
            0 = s(z') ]; % eqE0s
            0 = s(z') => 0 = s(z') ];
          0 = 0 => 0 = s(z') => 0 = s(z') ]; % case (z = s(z'))
        0 = 0 => 0 = z => 0 = z;
      !z:nat. 0 = 0 => 0 = z => 0 = z; % case (y = 0)
```

<sup>2</sup>Warning: the “proof” below has not been verified by a proof checker!

```

[ y' : nat; !z:nat 0 = y' => y' = z => 0 = z;
  [ z : nat;
    [ 0 = s(y');
      [ s(y') = z;
        0 = z ];
        % eqE0s
      s(y') = z => 0 = z ];
    0 = s(y') => s(y') = z => 0 = z ];
  !z:nat. 0 = s(y') => s(y') => z => 0 = z ]; % case (y = s(y'))

!z:nat. 0 = y => y = z => 0 = z ];
!y:nat. !z:nat. 0 = y => y = z => 0 = z ]; % base case (x = 0)

[ x' : nat, !y:nat. !z:nat. x' = y => y = z => x' = z % ind hyp (x)
  [ y : nat;
    [ z : nat;
      [ s(x') = 0;
        [ 0 = z;
          s(x') = z ];
          % eqEs0
        0 = z => s(x') = z ];
      s(x') = 0 => 0 = z => s(x') = z ];
      !z:nat. s(x') = 0 => 0 = z => s(x') = z; % case (y = 0)
      [ y' : nat, !z:nat. s(x') = y' => y' = z => s(x') = z;
        [ z : nat;
          [ s(x') = s(y');
            [ s(y') = 0;
              s(x') = 0 ];
              % eqEs0
            s(y') = 0 => s(x') = 0 ];
            s(x') = s(y') => s(y') = 0 => s(x') = 0; % case (z = 0)

          [ z' : nat, s(x') = s(y') => s(y') = z' => s(x') = z';
            [ s(x') = s(y');
              [ s(y') = s(z');
                x' = y'; % eqEss
                y' = z'; % eqEss
                !z:nat. x' = y' => y' = z => x' = z;
                x' = y' => y' = z' => x' = z';
                y' = z' => x' = z';
                x' = z';
                s(x') = s(z') ];
                % eqIs
              s(y') = s(z') => s(x') = s(z') ];
              s(x') = s(y') => s(y') = s(z') => s(x') = s(z') ];
              s(x') = s(y') => s(y') => z => s(x') = z ];
              !z:nat. s(x') = s(y') => s(y') = z => s(x') = z ];
              % case (y = s(y'))
            !z:nat. s(x') = y => y = z => s(x') = z ];
            !y:nat. !z:nat. s(x') = y => y = z => x = z ];
            % ind step (x = s(x'))
          !y:nat. !z:nat. x = y => y = z => x = z ];
        !x:nat. !y:nat. !z:nat. x = y => y = z => x = z;

```

Instead of giving the proof term in full, we give its specification. Recall that

$$trans : \forall x \in \mathbf{nat}. \forall y \in \mathbf{nat}. \forall z \in \mathbf{nat}. x =_N y \supset y =_N z \supset x =_N z$$

and therefore *trans* is a function of five arguments: natural numbers  $x$ ,  $y$ , and  $z$  and proof terms  $u : x =_N y$  and  $w : y =_N z$ . It has to return a proof term  $M : x =_N z$ . The proof above corresponds to the following specification.

$$\begin{array}{llll} trans & \mathbf{0} & \mathbf{0} & \mathbf{0} \quad u \quad w = \mathbf{eq}_0 \\ trans & \mathbf{0} & \mathbf{0} & (\mathbf{s}(z')) \quad u \quad w = \mathbf{eqE}_{0s}(w) \\ trans & \mathbf{0} & (\mathbf{s}(y')) & z \quad u \quad w = \mathbf{eqE}_{0s}(u) \\ trans & (\mathbf{s}(x')) & \mathbf{0} & z \quad u \quad w = \mathbf{eqE}_{s0}(u) \\ trans & (\mathbf{s}(x')) & (\mathbf{s}(y')) & \mathbf{0} \quad u \quad w = \mathbf{eqE}_{s0}(w) \\ trans & (\mathbf{s}(x')) & (\mathbf{s}(y')) & (\mathbf{s}(z')) \quad u \quad w = \\ & & & \mathbf{eq}_s(trans \ x' \ y' \ z' \ (\mathbf{eqE}_{ss}(u)) \ (\mathbf{eqE}_{ss}(w))) \end{array}$$

Note that all but the first and the last case are impossible, for which we provide evidence by applying the right elimination rule to either  $u$  or  $w$ . We can also see that the first argument to the recursive call to *trans* is at  $x'$  and the specification above therefore satisfies the restriction on primitive recursion. By comparing this to the formal proof (and also the omitted proof term) we can see the programming with equational specifications of this kind is much simpler and more concise than many other representations. There is ongoing research on directly verifying and compiling specifications, which is close to actual programming practice in languages such as ML or Haskell.

Symmetry of equality can be proven in a similar way. This proof and the corresponding specification and proof term are left as an exercise to the reader.

A second class of example moves us closer the extraction of functional programs on natural numbers from constructive proofs. Keeping in mind the constructive interpretation of the existential quantifier, how could we specify the predecessor operation? There are many possible answers to this. Here we would like express that the predecessor should only be applied to positive natural numbers.

$$\forall x \in \mathbf{nat}. \neg x =_N \mathbf{0} \supset \exists y \in \mathbf{nat}. \mathbf{s}(y) =_N x$$

We can prove this by cases on  $x$ . Formally, this takes the form of an induction in which the induction hypothesis is not used.

**Proof:** The proof proceeds by cases on  $x$ .

**Case:**  $x = \mathbf{0}$ . Then the assumption  $\neg 0 =_N \mathbf{0}$  is contradictory.

**Case:**  $x = \mathbf{s}(x')$ . Assume  $\neg \mathbf{s}(x') =_N \mathbf{0}$ . We have to show that  $\exists y \in \mathbf{nat}. \mathbf{s}(y) =_N \mathbf{s}(x')$ . This follows with the witness  $x'$  for  $y$  since  $\mathbf{s}(x') =_N \mathbf{s}(x')$  by reflexivity of equality.

□

Here is the same proof in the linear notation for natural deductions.

```

[ x : nat;
  [ ~ 0 = 0;
    0 = 0;
    F;
    ?y:nat. s(y) = 0 ];
  ~ 0 = 0 => ?y:nat. s(y) = 0;    % case (x = 0)
[ x' : nat; ~ x' = 0 => ?y:nat. s(y) = x';
  [ ~ s(x') = 0;
    !z:nat. z = z;                % reflexivity lemma
    s(x') = s(x');
    ?y:nat. s(y) = s(x') ]
  ~ s(x') = 0 => ?y:nat. s(y) = s(x') ]; % case (x = s(x'))
~ x = 0 => ?y:nat. s(y) = x ];
!x:nat. ~ x = 0 => ?y:nat. s(y) = x;

```

Next we give the equational specification of the function  $pred'$  corresponding to this proof. Note that the function takes two arguments:  $x$  and a proof  $u$  of  $\neg x =_N \mathbf{0}$ . It returns a pair consisting of a witness  $n$  and proof that  $\mathbf{s}(n) =_N x$ .

$$\begin{aligned}
pred' \quad \mathbf{0} \quad u &= \mathbf{abort}(u \mathbf{eq}_0) \\
pred' \quad (\mathbf{s}(x')) \quad u &= \langle x', refl(\mathbf{s}(x')) \rangle
\end{aligned}$$

Note that in the case of  $x = \mathbf{0}$  we do not explicitly construct a pair, but abort the computation, which may have any type. This specification can be written as a program rather directly.

$$\begin{aligned}
pred' &: \forall x \in \mathbf{nat}. \neg x =_N \mathbf{0} \supset \exists y \in \mathbf{nat}. \mathbf{s}(y) =_N x \\
pred' &= \lambda x \in \mathbf{nat}. \mathbf{rec} \ x \\
&\quad \mathbf{of} \ f(\mathbf{0}) \Rightarrow (\lambda u. \mathbf{abort}(u \mathbf{eq}_0)) \\
&\quad | \ f(\mathbf{s}(x')) \Rightarrow (\lambda u. \langle x', refl(\mathbf{s}(x')) \rangle)
\end{aligned}$$

If we erase the parts of this term that are concerned purely with propositions and leave only data types we obtain

$$\begin{aligned}
pred' &= \lambda x \in \mathbf{nat}. \mathbf{rec} \ x \\
&\quad \mathbf{of} \ f(\mathbf{0}) \Rightarrow \_ \\
&\quad | \ f(\mathbf{s}(x')) \Rightarrow x'
\end{aligned}$$

which is close to our earlier implementation of the predecessor. Erasing the **abort** clause from the impossible case has left a hole, which we denoted by  $\_$ . We return to a formal specification of this erasure process in the next section.

First, we discuss an alternative way to connect arithmetic to functional programming. This is to write the program first and prove its properties. Recall the definition of  $pred$ :

$$\begin{aligned}
pred &= \lambda x \in \mathbf{nat}. \mathbf{rec} \ x \\
&\quad \mathbf{of} \ f(\mathbf{0}) \Rightarrow \mathbf{0} \\
&\quad | \ f(\mathbf{s}(x')) \Rightarrow x'
\end{aligned}$$

Now we can prove that

$$\forall x \in \mathbf{nat}. \neg x =_N \mathbf{0} \supset \mathbf{s}(\mathit{pred}(x)) =_N x$$

**Proof:** The proof is by cases over  $x$ .

**Case:**  $x = 0$ . Then  $\neg \mathbf{0} =_N \mathbf{0}$  is contradictory.

**Case:**  $x = \mathbf{s}(x')$ . Then

$$\begin{aligned} & \mathbf{s}(\mathit{pred}(\mathbf{s}(x'))) \\ \implies & \mathbf{s}(\mathbf{rec} \ \mathbf{s}(x')) \\ & \quad \mathbf{of} \ f(\mathbf{0}) \Rightarrow \mathbf{0} \\ & \quad \quad | \ f(\mathbf{s}(x')) \Rightarrow x' \\ \implies & \mathbf{s}(x') \end{aligned}$$

and  $\mathbf{s}(x') =_N \mathbf{s}(x')$  by reflexivity.

□

This shows that we must be able to use the rules of computation when reasoning about functions. This is not a property particular to natural numbers, but we might have to reason about functions at arbitrary types or proofs of arbitrary propositions. Reduction therefore has to be an integral part of the type theory. We will use two rules of the form

$$\frac{\Gamma \vdash M : A \quad A \implies A' \quad \Gamma \vdash A \ \mathit{prop}}{\Gamma \vdash M : A'} \ \mathit{conv}$$

$$\frac{\Gamma \vdash M : A' \quad A \implies A' \quad \Gamma \vdash A \ \mathit{prop}}{\Gamma \vdash M : A} \ \mathit{conv}'$$

where  $A \implies A'$  allows the reduction of a term occurring in  $A$ . A unified form of this rule where  $A \iff A'$  allows an arbitrary number of reduction and expansion steps in both directions is called *type conversion*. While reduction generally preserves well-formedness (see Theorem 3.1), the converse does not. Generally, this is implied either from the premise or the conclusion, depending on whether we are reasoning backward or forward. Note that the conversion rules are “silent” in that the proof term  $M$  does not change.

In the formal proof, computations are omitted. They are carried out implicitly by the type checker. The question whether the resulting checking problem is decidable varies, depending on the underlying notion of computation. We return to this question when discussing the operational semantics in Section ??.

We close this section with the formal version of the proof above. Note the use of the conversion rule *conv'*.

```
[ x : nat;
  [ ~ 0 = 0; 0 = 0; F;
    s(pred(x)) = x ];
  ~ 0 = 0 => s(pred(x)) = x;    % case (x = 0)
  [ x' : nat, ~ x' = 0 => s(pred(x')) = x';
    [ ~ s(x') = 0;
      !z:nat. z = z;           % reflexivity lemma
      s(pred(s(x'))) = s(x') ]; % since pred(s(x')) ==> x'
    ~ s(x') = 0 => s(pred(s(x'))) = s(x') ];
    % case (x = s(x'))
  ~ x = 0 => s(pred(x)) = x ];
!x:nat. ~ x = 0 => s(pred(x)) = x;
```